
Abstract

Normalization for the simply-typed � -calculus is proven in Twelf, an implementation of the Edinb urgh
Logical Framework. Since due to proof-theoretical restric tions Twelf Tait's computability method does not
seem to be directly usable, a syntactical proof is adapted an d formalized instead. In this case study, some
boundaries of Twelf current capabilities are touched and di scussed.

1 Introduction

Twelf is an implementation of the Edinburgh Logical Framework which supports rea-
soning in full higher-order abstract syntax (HOAS); therefore it is an ideal candidate
for reasoning comfortably about properties of prototypical programming languages
with binding. Previous work has focused on properties like subject reduction, con-

uence, compiler correctness. Even cut elimination for various sequent calculi has
been proven successfully. But until recently, there were noformalized proofs of nor-
malization 2 in Twelf. The reason might be that normalization is typicall y proven
by Tait's method, which cannot be applied directly in Twelf. This work explains
why Tait's method is at least not directly applicable and provides a syntactical
proof for the simply-typed lambda-calculus.

1 Research supported by the Graduiertenkolleg Logik in der Informatik of the Deutsche Forschungsgemein-
schaft, the thematic networks TYPES (IST-1999-29001) and Applied Semantics II (IST-2001-38957) of the
European Union and the project CoVer of the Swedish Foundation of Strategic Research.
2 There have been normalization proofs in logical frameworks with inductive de�nitions, for instance,
Altenkirch's proof of strong normalization for System F in LEGO [ 2]. Since HOAS is not available in a
framework like LEGO, he represents terms using de Bruijn ind ices.

This is a preliminary version. The �nal version will be publi shed in
Electronic Notes in Theoretical Computer Science

URL: www.elsevier.nl/locate/entcs

K ::= type kind of types
j f X :AgK dependent function kind

A ::= F M1 : : : Mn base type
j f X :AgA dependent function type
j A ! A non-dependent function type

M ::= C term constant
j X term variable
j [X :A]M term abstraction
j M M term application

Fig. 1. Syntactic classes of LF.

2 Twelf

The Edinburgh Logical Framework (LF 3 ) [6,7] is a dependently-typed lambda-
calculus with type families and �� -equality, but neither polymorphism, inductive
data types nor recursion. Expressions are divided into three syntactic classes: kinds,
types and terms, generated by the grammar in Fig.1. Herein, the meta variable
X ranges over a countably in�nite set of variable identi�ers, while F resp. C range
over type-family resp. term constants provided in a signature �. Note that neither
a type nor a kind can depend on a type; consequently, abstraction is missing on the
type level [11, p. 1124].

The framework comes with judgements for typing, M : A, kinding, A : K, and
wellformedness of kinds,K kind, plus �� -equality on for terms, types, and kinds
[7]. An object theory can be described in the framework by providing a suitable
signature � which adds kinded type family constants F : K and typed term constants
C : A.

Twelf [12] is an implementation of LF whose most fundamental task is tocheck
typing (and kinding) of a user given signature �, usually pro vided as a set of ASCII
�les. Symbols reserved for the framework are the following.

: . ( ) [ ] { } -> type

All others can be used to denote entities in the object theories. In the remainder of
this section, we show how to represent the simply-typed� -calculus with weak head
reduction, as speci�ed in Fig. 2, in Twelf.

2.1 Representation of Syntactic Objects

Untyped lambda terms t can be represented by one type family constanttm and
two term constants:

tm : type.
lam : (tm -> tm) -> tm.
app : tm -> tm -> tm.

3 This is not to be confused with Martin-L•of's framework for d ependent type theory, which is also abbre-
viated by LF.

Syntax.

r; s; t; u ::= x j �x:t j r s untyped terms
A; B; C ::= � j A ! B simple types
� ::= � j � ; x :A contexts (all variables distinct)

Type assignment � ` t : A. (x :A) 2 �
� ` x : A

of var

� ; x :A ` t : B
� ` �x:t : A ! B

of lam
� ` r : A ! B � ` s : A

� ` r s : B
of app

Weak head reduction t �! w t0.

(�x:t ) s �! w [s=x]t
beta

r �! w r 0

r s �! w r 0s
appl

Fig. 2. Simply-typed � -calculus and weak head reduction.

The lack of a construct for variables is due to the use of HOAS:object variables are
represented by variables of the framework, e. g., in the codefor the twice function:

twice = lam [f:tm] lam [x:tm] app f (app f x).

A more detailed explanation of higher-order encodings has been given by Sch•ur-
mann [17, p. 20�]. Simple types A can be generated from a nullary constant* for
some base type and a binary constant=>, used in�x, for function type formation.

ty : type.
* : ty.
=> : ty -> ty -> ty. %infix right 10 =>.

2.2 Representation of Judgements and Relations

Type assignment for untyped terms, � ` t : A, can be represented by two constants
as well: one for function introduction and one for function elimination. Note that in
Twelf syntax, the types of new constants may contain free variables (capital letters),
which are regarded as universally quanti�ed on the outside.

of : tm -> ty -> type. %infix none 1 of.
of_lam : ({x:tm} x of A -> (T x) of B)

-> (lam [x:tm] T x) of (A => B).
of_app : R of (A => B) -> S of A -> (app R S) of B.

Again, there is no separate rule for the typing of variables,instead it is part of the
rule for abstraction. The premise of ruleof_lam is to be read as:

Consider a temporary extension of the signature by a fresh constant x:tm and
assumex of A. Then (T x) of B holds.

This adds adynamical typing rule x of A for each new variablex instead of inserting
a typing hypothesis x : A into the typing context �. Hence, we do not explicitely
encode �, but let the framework handle the typing hypotheses. The shape of such
dynamical rules can be declared in Twelf by a%worlds statement. In the case of
typing, this would be the following:

%block of_var : some {A:ty} block {x:tm}{p:x of A}.
%worlds (of_var) (of _ _).

This states that in of -derivations, only assumptions blocks that are of the form
x:tm,p:x of A for someA:ty enter the global Twelf context. Worlds declarations
are required for coverage checking (see below).

Similar to the typing relation, we can represent weak head reduction t �! w t0,
which eliminates the head (resp. key) redex in termt but does not step under a
binding.

-->w : tm -> tm -> type. %infix none 1 -->w.
beta : app (lam T) S -->w T S.
appl : R -->w R' -> app R S -->w app R' S.

One advantage of HOAS is that substitution does not have to bede�ned, but can be
inherited from the framework. Since in rulebeta , term T : tm -> tm is � -function,
substitution [ u=y]t is simply expressed as applicationT U.

Lemma 2.1 If t �! w t0 then [u=y]t �! w [u=y]t0.

Proof. By induction on the derivation of t �! w t0.

� Case (�x:t ) s �! w [s=x]t. W.l.o.g. x 6= y and x not free in u. Then,

[u=y](( �x:t ) s) = ( �x: [u=y]t) [u=y]s

�! w [[u=y]s=x][u=y]t = [ u=y][s=x]t:

� Caser s �! w r 0s with r �! w r 0. By ind. hyp., [ u=y]r �! w [u=y]r 0. Hence,

[u=y](r s) = ([ u=y]r ) ([u=y]s)

�! w ([u=y]r 0) ([u=y]s) = [ u=y](r 0s)

2

Fig. 3. Weak head reduction is closed under substitution.

2.3 Representation of Theorems and Proofs

Fig. 3 shows the �rst lemma of our object theory. How do we representit? Twelf's
internal logic is constructive, therefore the lemma must beinterpreted construc-
tively: Given a derivation P of t �! w t0 and a term u, we can construct a deriva-
tion P0 of [u=y]t �! w [u=y]t0. In type theories with inductive types and recursion,
like Agda, Coq [8] and LEGO [14], the lemma would be represented as a recursive

function of the dependent type

� t : tm: � t0: tm: � P : t �! w t0: � u : tm: � y :var : [u=y]t �! w [u=y]t0:

In Twelf, however, with no recursive functions at hand, the lemma is represented
as a relation between input and output derivations, and, thus, via the propositions-
as-types paradigm, as just another type family.

subst_red : {U:tm} ({y:tm} T y -->w T' y)
-> T U -->w T' U -> type.

%mode subst_red +U +P -P'.

The %modestatement marks the �rst two arguments of type family subst_red as
inputs (+) and the third as output ( � ). Thus, the lemma is a functional relation,
and its proof is a logic program with two clauses, one for eachcase in the proof.

subst_red_beta: subst_red U ([y] beta) beta.
subst_red_appl: subst_red U ([y] appl (P y)) (appl P')

<- subst_red U P P'.
%terminates P (subst_red _ P _).

The base case of the induction is given by the constantsubst_red_beta , and the
step case, which appeals to the induction hypothesis, bysubst_red_appl . The types
of these constants are the actual program and correspond to PROLOG clauses. Note
that in the second type a reversed arrow \<- ", which resembles PROLOG's \:- ",
has been used to encourage an operational reading:

Substitution in a derivation which can be split into last rul e appl and remainder
P | P may mention y | results in a derivation P' extended by an application of
rule appl . Herein, P' is constructed from P recursively.

Since it is a logic program, we can even \execute" the lemma. Execution in Twelf is
search: Given a type with free variables, �nd an inhabitant of the type and solutions
for the free variables. For example:

P : {y} app (app (lam [x] x) y) y -->w app y y
= [y] appl beta.

%define P' = X
%solve K : subst_red (lam [z] z) P X.

This de�nes a 2-rule derivation P which witnesses that (�x:x ) y y �! w y y. The
%solve statement asks Twelf for a derivation P' which arises fromPby substituting
�z:z for y, according to the lemma. The answer is:

P' : app (app (lam [x] x) (lam [z] z)) (lam [z] z)
-->w app (lam [z] z) (lam [z] z)

= appl beta.
K : subst_red (lam [z] z) P (appl beta)

= subst_red_appl subst_red_beta.

Since the value ofP' equals P, the shape of the derivation has not changed, only
its result: the type of P' . The value of K gives an execution trace of logic program
subst_red : First, clause subst_red_appl has �red, then clause subst_red_beta

has concluded the search.

2.4 External Properties: Termination and Coverage

A logic program in Twelf corresponds to a partial function from inputs to outputs
as speci�ed by the mode declaration. Since only total functions correspond to
valid inductive proofs we must ensure that the de�ned function terminates on all
inputs and covers all possible cases. Both properties cannot be shownwithin the
framework, e. g., we cannot give a proof thatsubst_red is terminating. Instead,
totality of a function needs external reasoning and can be ensured by built-in tactics.

Rohwedder and Pfenning [15] and Pientka [13] contributed a termination checker
which is invoked by the %terminates pragma. In the case ofsubst_red , the second
argument P decreases structurally in each recursive call. Case coverage is ensured
by an algorithm by Pfenning and Sch•urmann [18]. Both termination and coverage
checking are necessarily incomplete. But all proofs developed in the remainder of
this article pass the totality check.

3 A Formalized Proof of Weak Normalization

In this section, we present a combinatorial proof of weak normalization for the
simply-typed lambda-calculus. It is similar to the textbook proof in Girard, Lafont
and Taylor [4, Ch. 4], but we avoid reasoning with numbers altogether. In fact, we
follow closely the very syntactical presentation of Joachimski and Matthes [9], which
has also been implemented in Isabelle/Isar by Nipkow and Berghofer [3]. The main
obstacle to a direct formalization in Twelf is the use of a vector notation for terms
by Joachimski and Matthes, which allows them to reason on a high level in some
cases. In this section, we will see a \de-vectorized" version of their proof which can
be outlined as follows:

(i) De�ne an inductive relation t * A.

(ii) Prove that for every term t : A the relation t * A holds.

(iii) Show that every term in the relation is weakly normaliz ing.

3.1 Inductive Characterization of Weak Normalization

Inductive characterizations of normalization go back to Goguen [5] and van Raams-
donk and Severi [20] and seem to have been independently discovered by Valentini
[19]. However all these de�nitions use vectors of arguments for� -terms. To make
such a de�nition Twelf-digestible, we have found a vector-free formulation. We in-
troduce a relation � ` t * A which stipulates that t is weakly normalizing of type
A, and an auxiliary relation � ` t +x A which additionally claims that t = x s for

some sequence of normalizing termss, i.e., t is neutral resp. head-redex free.

(x :A) 2 �
� ` x +x A

wnevar
� ` r +x A ! B � ` s * A

� ` r s +x B
wneapp

� ` r +x A
� ` r * A

wn ne
� ; x :A ` t * B

� ` �x:t * A ! B
wn lam

r �! w r 0 � ` r 0 * A
� ` r * A

wn exp

The Twelf representation is similar to the typing relation: Again, � and the hy-
pothesis rule are indirectly represented in rulewn_lamand the worlds declaration.

var : tm -> type.
wne : tm -> ty -> var X -> type.
wn : tm -> ty -> type.

wne_app : wne R (A => B) V -> wn S A -> wne (app R S) B V.
wn_ne : wne R A V -> wn R A.
wn_lam : ({x:tm} {v:var x} wne x A v -> wn (T x) B)

-> wn (lam T) (A => B).
wn_exp : R -->w R' -> wn R' A -> wn R A.

%block wne_var : some{A:ty}block {x:tm}{v:var x}{d:wne x A v}.
%worlds (wne_var) (wne _ _ _) (wn _ _).

The judgement var X states that X is a variable of type tm, since there are rules to
introduce var . This formalization of term variable was suggested to me by Chung-
Chieh Shan and enables coverage checking.

3.2 Closure under Application and Substitution

To show that each typed term t : A is in the relation t * A, we will proceed by
induction on the typing derivation. Di�cult is the case for a n application of the form
(�x:r ) s. It can only be shown to be in the relation by rule wn_exp, which requires
us to prove that [s=x]r is in the relation. If x is head variable of r , substitution
might create new redexes. In this case, however, we can arguethat the type of r is
a smaller type than the one ofs. These preliminary thoughts lead to the following
lemma.

Lemma 3.1 (Application and Substitution) Let D :: � ` s * A.

(i) If E :: � ` r * A ! C then � ` r s * C.

(ii) If E :: � ; x :A ` t * C, then � ` [s=x]t * C.

(iii) If E :: � ; x :A ` t +x C, then � ` [s=x]t * C and C is a part of A.

(iv) If E :: � ; x :A ` t +y C with x 6= y, then � ` [s=x]t +y C.

In Twelf, the lemma is represented by four type families. Theinvariant that
C is a subexpression ofA will be expressed via a%reducesstatement later, which

makes is necessary to make typeC an explicit argument to type family subst_x .

app_wn : {A:ty} wn S A
-> wn R (A => C) -> wn (app R S) C -> type.

subst_wn : {A:ty} wn S A -> ({x:tm} {v:var x}
wne x A v -> wn (T x) C) -> wn (T S) C -> type.

subst_x : {A:ty} wn S A -> {C:ty} ({x:tm} {v:var x}
wne x A v -> wne (T x) C v) -> wn (T S) C -> type.

subst_y : {A:ty} wn S A -> ({x:tm} {v:var x}
wne x A v -> wne (T x) C Y) -> wne (T S) C Y -> type.

%mode app_wn +A +D +E -F.
%mode subst_wn +A +D +E -F.
%mode subst_x +A +D +C +E -F.
%mode subst_y +A +D +E -F.

Proof of Lemma 3.1 Simultaneously by main induction on type A and side in-
duction on the derivation E. (By this we mean lexicographic induction on (A; E).)

(i) Show � ` r s * C. If the last rule of E was wn_ne, hence,r is neutral, then r s
is also neutral by rule wne_app, thus, it is in the relation. If the last rule was
wn_exp, we can apply the side ind. hyp. The interesting case isr = �x:t and

� ; x :A ` t * C
� ` �x:t * A ! C

wn lam:

Here, we proceed by side ind. hyp.ii . Here is the corresponding Twelf code (we
repeat the type of app_wn).

app_wn : {A:ty} wn S A
-> wn R (A => C) -> wn (app R S) C -> type.

app_wn_ne : app_wn A D (wn_ne E) (wn_ne (wne_app E D)).
app_wn_exp : app_wn A D (wn_exp P E) (wn_exp (appl P) F)

<- app_wn A D E F.
app_wn_lam : app_wn A D (wn_lam E) (wn_exp beta F)

<- subst_wn A D E F.

(ii) Show � ` [s=x]t * C for � ; x : A ` t * C. If t is not neutral, we conclude
by ind. hyp. and possibly Lemma 2.1. Otherwise, we distinguish on the head
variable of t: is it x, then we proceed by side ind. hyp.iii , otherwise by side
ind. hyp. iv.

subst_wn : {A:ty} wn S A -> ({x:tm} {v:var x} wne x A v
-> wn (T x) C) -> wn (T S) C -> type.

subst_wn_lam: subst_wn A D ([x][vx][dx]
wn_lam ([y][vy][dy] E y vy dy x vx dx)) (wn_lam F)

<- {y}{vy}{dy} subst_wn A D (E y vy dy) (F y vy dy).

subst_wn_exp: subst_wn A (D : wn S A) ([x][v][dx]
wn_exp (P x) (E x v dx)) (wn_exp P' E')

<- subst_wn A D E E'
<- subst_red S P P'.

subst_wn_x : subst_wn A D ([x][v][dx]
(wn_ne (E x v dx) : wn (T x) C)) F

<- subst_x A D C E F.

subst_wn_y : subst_wn A D ([x][v][dx]
wn_ne (E x v dx)) (wn_ne F)

<- subst_y A D E F.

(iii) Show � ` [s=x]t * C for � 0 ` t +x C with � 0 := � ; x : A. In case t = x, the
type C is trivially a part of A = C and we conclude by assumption � ` s * C.
Otherwise, t = r u and the last rule in E was

� 0 ` r +x B ! C � 0 ` u * B
� 0 ` r u +x C

wneapp:

By side ind. hyp. iii we know that B ! C is a part of A and � ` r 0 * B ! C
where r 0 := [ s=x]r . Similarly � ` u0 * B for u0 := [ s=x]u by side ind. hyp. ii .
Since B is a strict part of A, we can apply the main ind. hyp. i and obtain
� ` r 0u0 * C.

subst_x : {A:ty} wn S A -> {C} ({x:tm} {v:var x} wne x A v
-> wne (T x) C v) -> wn (T S) C -> type.

subst_x_x : subst_x A D A ([x][v][dx]dx) D.
subst_x_app : subst_x A D C ([x][v][dx]

wne_app (E x v dx)
(F x v dx : wn (S x) B)) EF

<- subst_x A D (B => C) E E'
<- subst_wn A D F F'
<- app_wn B F' E' EF.

%reduces C <= A (subst_x A D C E F).

The %reduces declaration states that the type expressionC is a subexpres-
sion of A. Twelf checks that this invariant is preserved in all possibilities of
introducing subst_x A D C E F. In casesubst_x_x it holds becauseC is in-
stantiated to A. In casesubst_x_app it follows from the ind. hyp. which states
that already B => Cis a subexpression ofA.

(iv) Show � ` [s=x]t +y C for � ; x : A ` t +y C. There a two cases.t = y, which
holds immediately, and t = r u , which follows from side ind. hyp.s ii and iv.
In our Twelf representation, we cannot distinguish variable y from any other
term, so we widen the �rst case to cover allt such that x is not free in t. This

is expressed by lettingE not refer to x or dx.

subst_y : {A:ty} wn S A -> ({x:tm} {v:var x} wne x A v
-> wne (T x) C Y) -> wne (T S) C Y -> type.

subst_y_y : subst_y A D ([x][v][dx] E) E.
subst_y_app : subst_y A D ([x][v][dx]

wne_app (E x v dx) (F x v dx)) (wne_app E' F')
<- subst_y A D E E'
<- subst_wn A D F F'.

2
To justify the appeals to the ind. hyp.s we invoke the Twelf termination checker
with the following termination order.

%terminates {(Ax Ay As Aa) (Ex Ey Es Ea)}
(subst_x Ax _ _ Ex _)
(subst_y Ay _ Ey _)
(subst_wn As _ Es _)
(app_wn Aa _ Ea _).

It expresses that the four type families are mutually recursive and terminate w. r. t.
the lexicographic order on pairs (A; E) of types A and typing derivations E. This
corresponds on a main induction onA and a side induction onE. To verify termina-
tion, Twelf makes use of the%reducesdeclaration. Using the world block wne_var,
the code also coverage-checks, hence, the lemma is completely machine-veri�ed.

Lemma 3.2 If D :: � ` t : A then � ` t * A.

Proof. By induction on D, using the application lemma. Can be directly formalized
in Twelf [ 1]. 2

3.3 Soundness of Inductive Characterization

To complete our proof of weak normalization, we need to show that for each term
t in the relation t * A or t +x A there exists a normal form v such that t �! �

v. After formulating full � -reduction �! with the usual closure properties, the
proof is a simple induction on the derivation E :: t * A resp. E :: t +x A. The
formalization of multi-step � -reduction �! � and fundamental properties like closure
under substitution is completely standard and available online [1]. Similar proofs
can be found in Sch•urmann's thesis [17, Ch. 3].

Typed � -normal forms can be de�ned inductively by two judgements � ` t " A,
t of type A is normal, and � ` t # A, t of type A is normal and neutral, the last
adjective meaning that t = x s for somes.

(x :A) 2 �
� ` x # A

ne var
� ` r # A ! B � ` s " A

� ` r s # B
ne app

� ` r # A
� ` r " A

nf ne
� ; x :A ` t " B

� ` �xt " A ! B
nf lam

Lemma 3.3 (Soundness)

(i) If D :: � ` t +x A then � ` t0 # A for some t0 � � t .

(ii) If D :: � ` t * A then � ` t0 " A for some t0 � � t .

Proof. Simultaneously by induction on D. The Twelf formalization is unproblem-
atic. 2

Theorem 3.4 (Weak normalization) If � ` t : A then � ` t0 " A for some
t0 � � t .

The theorem is the just the composition of lemmata3.2 and 3.3.

Remark 3.5 � -long � -normal forms can be enforced by requiringA to be a base
type in rule nf ne. The expansion of a� -normal term t to an � -long form can be
de�ned by induction on � ` t " A. Thus, our judgement t * A captures trivially
also the weakly �� -normalizable terms.

4 On Proof-Theoretical Limitations of Twelf

Having successfully completed the proof of weak normalization we are interested
whether it could be extended to strong normalization and stronger object theo-
ries, like G•odel's T. In this section, we touch these questions, but our answers are
speculative and preliminary.

Joachimski and Matthes [9] extend their proof to G•odel's T, using the in�ni-
tary ! -rule to state when a recursive function over natural numbers is weakly nor-
malizing. Their proof is not directly transferable since only �nitary rules can be
represented in Twelf.

For the same reason, Tait's proof cannot be formalized directly in Twelf. Its key
part is the de�nition

8s: s * A ) r s * B
r * A ! B

with an in�nitary premise. Its literal translation into Twe lf

wn_arr : ({S:tm} wn S A -> wn (app R S) B) -> wn R (A => B)

means something else, namely \if for a fresh termS for which we assumewn S A
it holds that wn (app R S) B, then wn R (A => B)". Translating this back into
mathematical language, we obtain the rule

x * A ) r x * B
r * A ! B

for a fresh variable x:

Since variablesx are weakly normalizing anyway, we can simplify the premise further
to r x * B , obtaining clearly something we did not want in the �rst plac e.

Recently, Sarnat and Sch•urmann [16] have proven weak normalization for the
simply-typed � -calculus in Twelf using logical relations. They �rst give a inductive
characterization of weakly normalizing terms similar to ours. Instead of directly
proving that it is closed under application (our Lemma 3.1) they follow a Tait-
style logical relation argument. To make it work, they de�ne minimal �rst-order

logic (MFOL) and prove cut elimination. The logical relatio n is then represented
within MFOL, but they can show an escape lemma: each term which is provably in
the logical relation is actually weakly normalizing. The proof complexity has been
shifted to cut elimination; it is established by a lexicographic induction just as our
application/substitution lemma.

Sarnat and Sch•urmann have formalized the �rst logical relations proof in Twelf,
an impressive achievement, since it was unknown for a long time how to do this.
Yet, we have given a much simpler proof of the same result. It is unclear whether
their approach scales to stronger systems, e. g., G•odel's T. At least, the termination
checker of Twelf must be extended to lift the proof-theoretic strength of Twelf
beyond primitive recursive arithmetic.

5 Strong Normalization

Due to the interpretation of quanti�cation in Twelf, in�nit ary rules cannot be rep-
resented, which also obstructs the de�nition of the predicate strongly normalizing
sn by the inductive rule

8t0: t �! t0 ) snt0

snt
;

expressing that the set of strongly normalizing terms is theaccessible part of the
reduction relation.

However, the inductive characterization of weakly normalizing terms can be
modi�ed to account for strongly normalizing terms. A term ( �xt ) s is weakly nor-
malizing i� [ s=x]t is|this holds for strong normalization if we additionally r equire
that s is strongly normalizing. Thus, in each weak head expansion step, the term
s must be strongly normalizing. We inductively de�ne such a expansion relation
�! sn together with the set of strongly normalizing terms.

(x :A) 2 �
� ` x +x A

sne var
� ` r +x A ! B � ` s * A

� ` r s +x B
sne app

� ` r +x A
� ` r * A

sn ne
� ; x :A ` t * B

� ` �x:t * A ! B
sn lam

� ` r �! sn r 0 � ` r 0 * A
� ` r * A

sn exp

� ` s * A
� ` (�xt ) s �! sn [s=x]t

sr beta
� ` r �! sn r 0

� ` r s �! sn r 0s
sr app

The soundness of the inductive characterization must be established on paper,
but an application/substitution lemma can now be proved in T welf analogously
to Lemma 3.1, from which it follows that each well-typed term is strongly normal-
izing.

6 Conclusions

We have presented a formalization of Joachimski and Matthes' version of an ele-
mentary proof of weak normalization of the simply-typed � -calculus in Twelf. We
further have outlined a partial formalization of strong nor malization proof and dis-
cussed feasibility of Tait-style proofs in Twelf.

Related work: Hereditary substitutions.
Watkins et al. [21] have formulated a concurrent logical framework where all

terms are in normal form. Consequently, substitution and application must per-
form some evaluation. Watkin's substitution and applicati on algorithms, called
hereditary substitutions, match our Lemma 3.1 precisely. Recently, Lee, Crary,
and Harper [10] have formalized a full intermediate language for SML basedon
hereditary substitutions.

